

序

Excel 是各行各業極為常用的試算表軟體，具備製作表格、計算、資料分析與統計等豐富功能。其中，與計算功能有關的函數多達 480 個以上。

這是一本 Excel 函數百科，網羅了完整的函數。書中以淺顯易懂的說明，介紹函數的功能與用法，並視狀況列舉使用範例。針對使用頻率較高的函數，特別準備了多個範例，說明與其他函數一起使用時的運用方法。本書提供了使用範例的電子檔，你可以自行下載，這些檔案應能幫助你實際運用函數並加深理解。

另外，書中透過專欄與提示，解說與函數有關的知識，可以讓你瞭解較難的用語和內容。

除了函數之外，也整理了使用函數時，必須具備的基本知識。即便你是初學者，也能順利運用這些函數。

除此之外，書中還介紹了方便運用函數的進階技巧，以及挑選函數時，應先瞭解的實用功能與靈活運用函數的方法。

本書依照類別、英文字母、目的等分類準備了三種索引，當你在搜尋函數時，請善加運用。

如果這本書能幫助所有使用 Excel 的人有效提升技能及改善工作效率，筆者將深感榮幸。

最後，筆者在此由衷感謝編寫本書時，提供協助的所有人員。

国本 温子

- 以下是本書的版面結構範例。
請注意！本書會依照 Excel 函數的使用頻率、使用難度，調整解說內容的份量。

■ Excel 函數支援的版本
支援版本 365

■ Excel 函數的類別 / 子類別

- 依 Excel 函數的性質、用途分類
- Excel 函數的基本資料
函數名稱
函數的用法
函數的說明
- 函數的格式
說明函數的格式
- 函數的引數
引數的說明
- 函數的使用範例
介紹實際的用法與搭配其他函數的範例
- 相關內容
提供搭配使用會更方便的函數及技巧

數學與三角函數

亂數

365
2019
2016
2013

RANDARRAY

建立含亂數的陣列表格

傳回隨機數值的陣列。開啟活頁簿，按下 [F9] 鍵，每次重新計算工作表時，就會產生新的亂數。

格式： **RANDARRAY([列數],[欄數],[最小值],[最大值],[整數])**

- 在 [列數]×[欄數] 的陣列中，於 [最小值] 與 [最大值] 的範圍內，在 [整數] 設定整數或小数，建立亂數清單。
- 在 [列數] 設定陣列的列數。
- 在 [欄數] 設定陣列的欄數。
- 在 [最小值] 設定亂數的最小值。省略時，預設為 0。
- 在 [最大值] 設定亂數的最大值。省略時，預設為 1。
- 在 [整數] 設定是否將亂數變成整數。TRUE 是整數，FALSE 或省略時是實數亂數。

範例 ① 建立 3 列 3 欄由 10 到 100 的亂數表

	A	B	C	D	E	F	G	H	I
1	亂數範例表				亂數範例表				
2	1	2	3	1	2	3			
3	1	70	57	1	10.372	66.538	10.938		
4	2	80	34	95	2	65.797	89.412	54.541	
5	3	40	31	52	3	15.49	76.736	77.965	

公式 =RANDARRAY(3,3,10,100,TRUE)

公式 =RANDARRAY(3,3,10,100,FALSE)

說明 以最小值為 10，最大值為 100 建立 3 列 × 3 欄的亂數陣列表格。B3 儲存格的第 5 引數 [整數] 為 True，所以產生整數亂數，G3 儲存格為 FALSE，因而產生小數亂數。在 B3 儲存格、G3 儲存格輸入 RANDARRAY 函數，按下 [Enter] 鍵，確定公式後，會自動以陣列顯示結果。Microsoft 365 是利用溢出功能，自動輸入必要的動態陣列公式。

相關

RAND 以大於 0 小於 1 的實數產生亂數 → p.73

RANDBETWEEN 產生整數亂數 → p.74

DATE

從年、月、日取得日期

組合年、月、日，傳回日期的序列值。

格式： **DATE(年, 月, 日)**

- 在 [年] 設定 0 ~ 9999 之間的整數。如果是 0 ~ 1899，就加上 1900。例如「=DATE(21,3,3)」的年份是「1921」(1900+21)，傳回「1921/3/3」。若是 1900 ~ 9999，該值就是實際的年份。
- 在 [月] 設定 1 ~ 12 之間的整數。如果設定了超出這個範圍的數值，會自動調整日期。例如「=DATE(2020,13,1)」是 12 月的一個月後，變成「2021/1/1」，而「=DATE(2020,0,1)」是 1 月的一個月前，因此傳回「2019/12/1」。
- 在 [日] 設定 1 ~ 31 之間的整數。如果設定了超出這個範圍的數值，會自動調整日期。例如「=DATE(2020/4/0)」是 1 日的前一天，變成上個月的月底「2020/3/31」。

範例 1 從出生年月日取出年、月、日，顯示今年的生日

	A	B	C	D	E	F	G	H	I
1	會員姓名	出生年月日	年	月	日	今年的生日			
2	井上花子	1996/6/9	1996	6	9	2022/6/9	公式	=DATE(YEAR(NOW()),D2,E2)	
3									
4									

公式 =YEAR(B2)

公式 =MONTH(B2)

公式 =DAY(B2)

說明 根據 B2 儲存格輸入的日期，在 C2 儲存格使用 YEAR 函數，在 D2 儲存格使用 MONTH 函數，在 E2 儲存格使用 DAY 函數，分別取出年、月、日。在 F2 儲存格使用 DATE 函數，利用「YEAR(NOW())」，從現在的日期取出今年的年份，在 D2 儲存格設定月，E2 設定日，計算出今年的生日。

日期和時間

時間

365

2019

2016

2013

HOUR

從時間取得小時數

從時間取得小時數。傳回值是 0 ~ 23 之間的整數。

格式： HOUR(序列值)

在 [序列值] 以序列值或 "8:45 AM" 這樣的字串設定時間。

參照在儲存格內輸入的時間，或使用 TIME 函數、TIMEVALUE 函數等，可以設定時間的序列值。

日期和時間

時間

365

2019

2016

2013

MINUTE

從時間取得分鐘數

從時間取得分鐘數。傳回值是 0 ~ 59 之間的整數。

格式： MINUTE(序列)

在 [序列值] 以序列值或 "8:45 AM" 這樣的字串設定時間。

參照在儲存格內輸入的時間，或使用 TIME 函數、TIMEVALUE 函數等，可以設定時間的序列值。

日期和時間

時間

365

2019

2016

2013

SECOND

從時間取得秒數

從時間取得秒數。傳回值是 0 ~ 59 之間的整數。

格式： SECOND(序列值)

在 [序列值] 以序列值或 "8:45:15 AM" 這樣的字串設定時間。

參照在儲存格內輸入的時間，或使用 TIME 函數、TIMEVALUE 函數等，可以設定時間的序列值。

相關

TIME 從時、分、秒取得時間 → p.84

TIMEVALUE 把代表時間的字串轉換成序列值 → p.97

TIME

從時、分、秒取得時間

組合時、分、秒傳回時間的序列值。傳回值為序列值，是 0 (0:00:00) ~ 0.99988426 (23:59:59) 之間的小數。

格式：**TIME(時, 分, 秒)**

- 在 [時] 設定 0 ~ 32767 之間的數值。如果設定了 24 以上的數值，會將除以 24 後的餘數設定成時間。例如「=TIME(30,0,0)」會視為「=TIME(6,0,0)」，傳回「6:00 AM」(序列值：0.25)。
- 在 [分] 設定 0 ~ 32767 之間的數值。如果設定了 60 以上的數值，將轉換成時與分。例如「=TIME(0,90,0)」會視為「=TIME(1,30,0)」，傳回「1:30 AM」(序列值：0.0625)。
- 在 [秒] 設定 0 ~ 32767 之間的數值。如果設定了 60 以上的數值，將轉換成時、分、秒。例如「=TIME(0,0,1800)」會視為「=TIME(0,30,0)」，傳回「12:30 AM」(序列值：0.020833333)。

範例 1 計算扣除休息時間後的實質時間

	A	B	C	D	E	F	G	H	I	J
1	選手編號	時間	時	分	秒	扣除休息時間(15)				
2	101	03:45:25	3	45	25	03:30:25	公式 =B2-TIME(0,15,0)			
3										
4										
5										

公式 =HOUR(B2)

公式 =MINUTE(B2)

公式 =SECOND(B2)

說明 根據 B2 儲存格輸入的時間，在 C2 儲存格使用 HOUR 函數，在 D2 儲存格使用 MINUTE 函數，在 E2 儲存格使用 SECOND 函數，分別取出時、分、秒。在 F2 儲存格使用 TIME 函數，將 B2 儲存格的時間減去 15 分鐘，計算出扣除休息時間後的實質時間。

EDATE

計算指定月數前或後的日期

從開始日期起算，傳回與指定月數前、後的日期對應的序列值。例如，從生產日期或保固期間計算保存期限、每月定期購買的配送日。

格式：**EDATE(開始日期, 月)**

- 在 [開始日期] 設定起算日的日期。日期可以設定成序列值、日期字串 (如 "2022/5/20")、儲存格內輸入的日期參照、使用了 DATE 函數的日期序列值。
- 在 [月] 設定正值，會傳回 [開始日期] 之前的日期。若設定成負值，會傳回 [開始日期] 之後的日期。

範例 1 建立定期購買的每月配送日清單

	A	B	C	D	E	F	G	H
1	配送次數	日期						
2	1	1月15日						
3	2	2月15日						
4	3	3月15日						
5	4	4月15日						
6	5	5月15日						
7	6	6月15日						
8								
9								

公式 =EDATE(B2,1)

說明 把 B2 儲存格設定為開始日期，計算一個月後的日期。

TEXT

設定數值的顯示格式並轉換成字串

將數值設定成指定的顯示格式並轉換成字串。

格式： **TEXT(數值, 顯示格式)**

- 在 [數值] 設定要指定顯示格式的數值。
- 在 [顯示格式] 使用格式符號，以字串設定顯示格式。例如用 " " 包圍格式符號進行設定，如 " #,##0 " 。

範例 1 利用日期顯示星期名稱

	A	B	C
1	日期	星期1	星期2
2	2021/2/15	星期一	Monday
3	2021/2/16	星期二	Tuesday

公式 =TEXT(A2,"aaaa")

公式 =TEXT(A2,"dddd")

說明 在 B2 儲存格以「aaaa」格式顯示 A2 儲存格的日期是星期幾。在 C2 儲存格以「dddd」格式顯示星期名稱。

範例 2 連接日期、金額及字串

	A	B	C
1	日期	金額	連接文字
2	2021/2/15	\$15,000	2021/2/15的帳單：\$15,000。
3	2021/2/16	\$9,000	2021/2/16的帳單：\$9,000。
4	2021/2/17	\$28,000	2021/2/17的帳單：\$28,000。

公式 =CONCAT(TEXT(A2,"yyyy/m/d"),"的帳單:",TEXT(B2,"\$#,##0"),"。")

說明 使用 TEXT 函數設定日期(A2)與金額(B2)的顯示格式，並轉換成字串，再以 CONCAT 函數連接日期、金額及字串。

文字

轉換字串

365

2019

2016

2013

ASC

將全形文字轉換成半形

將全形英數文字（兩個位元組）轉換成半形英數文字（一個位元組）。

格式： **=ASC(字串)**

在 [字串] 設定想轉換成半形，包含英數文字的字串。字串內的中文字不會變成半形。例如「=ASC(" Excel 函數課程 ")」傳回「Excel 函數課程」。

文字

轉換字串

365

2019

2016

2013

BIG5

把半形文字轉換成全形

將半形英數文字（一個位元組）轉換成全形英數文字（兩個位元組）。

格式： **=BIG5(字串)**

在 [字串] 設定想轉換成全形，包含英數文字的字串。例如「=BIG5("Excel 基礎 2019")」傳回「Excel 基礎 2019」。

文字

轉換字串

365

2019

2016

2013

ROMAN

將數值轉換成羅馬數字

把數值轉換成以羅馬數字顯示的字串，如「I、II、III」。設定格式就能指定羅馬字的形式。

格式： **ROMAN(數值,[格式])**

- 在 [數值] 設定原始數值。
- 在 [格式] 設定羅馬數字的格式（請參考下表）。

格式

格式	種類
0,TRUE,省略	古典
1	比 0 還精簡的格式
2	比 1 還精簡的格式
3	比 2 還精簡的格式
4,FALSE	簡化（最簡化的格式）

文字

轉換字串

365

2019

2016

2013

ARABIC

把羅馬數字轉換成數值

把指定的羅馬數字轉換成數值(阿拉伯數字)。

格式： **ARABIC(字串)**

在 [字串] 設定羅馬數字。必須使用半形英文小寫或大寫設定。

文字

轉換字串

365

2019

2016

2013

YEN

將數值轉換成日元貨幣字串

將數值四捨五入至指定位數，使用日元貨幣格式(¥)轉換成字串。

格式： **YEN(數值, [位數])**

- 在 [數值] 設定原始數值。例如「=YEN(1500)」傳回「¥1,500」。
- 在 [位數] 設定要顯示的位數。例如設定為「2」，會四捨五入至小數點以下第二位。設定方法和 ROUND 函數一樣。省略時，預設為 0。

文字

轉換字串

365

2019

2016

2013

DOLLAR

將數值轉換成美元貨幣字串

將數值四捨五入至指定的位數，使用美元貨幣格式(\$)轉換成字串。

格式： **DOLLAR (數值, [位數])**

- 在 [數值] 設定原始數值。例如「= DOLLAR(1500)」傳回「\$1,500.00」。
- 在 [位數] 設定要顯示的位數。例如設定為「2」，會四捨五入至小數點以下第二位。設定方法和 ROUND 函數一樣。省略時，預設為 0。

文字

轉換字串

365

2019

2016

2013

▶ BAHTTEXT

將數值轉換成泰銖貨幣字串

把數值轉換成泰文，並加上代表 baht 的接尾字串。

格式： **BAHTTEXT(數值)**

在 [數值] 設定原始數值。例如「=BAHTTEXT(1500)」傳回「หนึ่งพันห้าร้อยบาทถ้วน」。

文字

轉換字串

365

2019

2016

2013

▶ LOWER

把英文轉換成小寫

把字串內的英文大寫全都轉換成小寫。

格式： **LOWER(字串)**

在 [字串] 設定原始字串。英文不論全形或半形都會進行轉換。例如「=LOWER("Apple pie")」傳回「apple pie」。

文字

轉換字串

365

2019

2016

2013

▶ UPPER

將英文轉換成大寫

把字串內的英文小寫全都轉換成大寫。

格式： **UPPER(字串)**

在 [字串] 設定原始字串。英文不論全形或半形都會進行轉換。例如「=UPPER("Apple pie")」傳回「APPLE PIE」。

文字

轉換字串

365

2019

2016

2013

PROPER

只將英文單字的第一個字母轉換成大寫

把字串內的英文單字第一個字母或符號後面的字母轉換成大寫，其餘英文字都轉換成小寫。

格式：**PROPER(字串)**

在 [字串] 設定原始字串。英文不論全形或半形都會進行轉換。例如「=PROPER("Apple pie")」傳回「Apple Pie」。

文字

轉換字串

365

2019

2016

2013

NUMBERSTRING

將數值轉換成國字

以指定格式將數值轉換成國字數字。

格式：**NUMBERSTRING(數值, 形式)**

- 在 [數值] 設定原始數值。
- 在 [格式] 以 1 ~ 3 的整數設定要轉換的國字格式(請參考下表)。例如「=NUMBERSTRING(12345,1)」傳回「一萬二千三百四十五」。

數值格式

格式	轉換文字 (以 15432 為例)	對應儲存格的顯示格式
1	一萬五千四百三十二	[DBNum1]
2	壹萬伍仟肆佰參拾貳	[DBNum2]
3	一五四三二	[DBNum1]#

Hint 這個函數無法從函數庫中選取，必須手動輸入公式。



財務函數



財務函數提供了可以計算每月貸款還款金額、以零存整付達到目標金額的儲蓄次數，以及固定利率債券的殖利率、應計利息等與投資相關的各種函數。

PMT

計算定期償還的貸款或儲蓄金額

傳回在固定利率及期間內，貸款為本金平均攤還的定期支付額，或零存整付的定期存款額。

格式：**PMT(利率, 期間, 現值, [未來值], [支付日期])**

- 在 [利率] 設定利率。若為每月支付，設定年利率 $\div 12$ 。
- 在 [期間] 設定總支付次數。若為每月支付，設定為年數 $\times 12$ 。
- [現值] 若是貸款，設定為借款金額；如果是儲蓄，則設定為頭期款。
- [未來值] 若為貸款，設定支付後的餘額（還清時為 0）；如果是儲蓄，設定最終的目標金額。省略時，預設為 0。
- [支付日期] 若為期初是 1，期末是 0。省略時，預設為 0。

範例 1 計算每月貸款的還款金額

	A	B	C
1	償還貸款		
2	借款	\$200,000	
3	利率 (年)	4.0%	
4	期間 (月)	12	
5	每月還款金額	\$-17,030	
6			

公式 = PMT(B3/12, B4, B2, 0, 0)

說明 假設年利率 4% (B3/12)、還款期間 12 個月 (B4)、借款 200,000 元 (B2) 時，計算每月的支付金額。定期還款以月為單位，所以年利率要除以 12，變成月利率。

相關

PPMT 計算貸款還款金額中攤還的本金 → p.282

IPMT 計算貸款還款金額中攤還的利息 → p.283


COLUMN

財務函數以「-」（負）代表支付，以「+」（正）表示取得。PMT 函數是用來計算支付金額，所以結果為「-」。假如不想顯示為負值，可以在函數前面加上「-」，就能反轉符號。

財務函數的期間與利率單位要與定期支付的單位一致。例如，定期支付以月為單位時，要將利率換算成月利率，期間換算成月數。

COLUMN

償還貸款的方法包括本息平均攤還與本金平均攤還。本息平均攤還是每次還款額固定，只改變本金與利息占還款金額的比例。本金平均攤還是在還款期間內，平均除以本金，根據餘額計算利息。因此初期還款金額較多，之後會逐漸減少。



PPMT

計算貸款還款金額中攤還的本金

以本金平均攤還的方式償還貸款時，傳回指定次數的付款金額中包含的本金。

格式：**PPMT(利率, 期, 期間, 現值, [未來值], [支付日期])**

- 在 [利率] 設定利率。若為每月支付，設定年利率 ÷ 12。
- 在 [期] 以介於 1 ~ [期間] 的範圍設定是第幾次付款。
- 在 [期間] 設定總支付次數。若為每月支付，設定為年數 × 12。
- [現值] 若是貸款，設定為借款金額；如果是儲蓄，則設定為頭期款。
- [未來值] 若為貸款，設定支付後的餘額（還清時為 0）；如果是儲蓄，設定最終的目標金額。省略時，預設為 0。
- [支付日期] 若為期初是 1，期末是 0。省略時，預設為 0。

範例 1 計算各還款次數的付款金額中包含的本金

	A	B	C	D	E
1	償還貸款			次數	攤還本金
2	借款	\$200,000		1	\$-16,363
3	利率 (年)	4.0%		2	\$-16,418
4	期間 (月)	12		3	\$-16,473
5	每月還款金額	\$-17,030		4	\$-16,527
6				5	\$-16,583
7				6	\$-16,638
8				7	\$-16,693
9				8	\$-16,749
10				9	\$-16,805
11				10	\$-16,861
12				11	\$-16,917
13				12	\$-16,973
14					

說明 假設年利率 4% (B3/12)、還款期間 12 個月 (B4)、借款 200,000 元 (B2) 時，計算每次還款 (D2) 的付款金額中所包含的本金。定期還款以月為單位，所以年利率要除以 12，變成月利率。

公式 =PPMT(\$B\$3/12,D2,\$B\$4,\$B\$2)

IPMT

計算貸款還款金額中攤還的利息

以本息平均攤還的方式償還貸款時，傳回指定次數的付款金額中包含的利息。

格式：**IPMT(利率, 期, 期間, 現值, [未來值], [支付日期])**

- 在 [利率] 設定利率。若為每月支付，設定年利率 ÷ 12。
- 在 [期] 以介於 1 ~ [期間] 的範圍設定是第幾次付款。
- 在 [期間] 設定總支付次數。若為每月支付，設定為年數 × 12。
- [現值] 若是貸款，設定為借款金額；如果是儲蓄，則設定為頭期款。
- [未來值] 若為貸款，設定支付後的餘額（還清時為 0）；如果是儲蓄，設定最終的目標金額。省略時，預設為 0。
- [支付日期] 若為期初是 1，期末是 0。省略時，預設為 0。

Hint PMT 函數(定期付款金額) = PPMT 函數(攤還本金) + IPMT 函數(攤還利息)的關係成立。

範例 1 計算各還款次數的付款金額中包含的利息

	A	B	C	D	E
1	償還貸款			次數	攤還利息
2	借款	\$200,000		1	\$-667
3	利率 (年)	4.0%		2	\$-612
4	期間 (月)	12		3	\$-557
5	每月還款金額	\$-17,030		4	\$-502
6				5	\$-447
7				6	\$-392
8				7	\$-337
9				8	\$-281
10				9	\$-225
11				10	\$-169
12				11	\$-113
13				12	\$-57
14					

公式 =IPMT(\$B\$3/12,D2, \$B\$4,\$B\$2)

說明 假設年利率 4 % (B3/12)、還款期間 12 個月(B4)、借款 200,000 元(B2) 時，計算每次還款(D2)的付款金額中包含的利息。定期還款以月為單位，所以年利率要除以 12，變成月利率。

CUMIPMT

計算貸款還款金額中累計支付的利息

以本息平均攤還的方式償還貸款時，傳回在指定期間內，累計支付的利息。

格式：**CUMIPMT(利率, 期間, 現值, 期初, 期末, 支付日期)**

- 在 [利率] 設定利率。若為每月支付，設定年利率 ÷ 12。
- 在 [期間] 設定總支付次數。若為每月支付，設定為年數 × 12。
- 在 [現值] 設定借款金額。
- 在 [期初] 設定計算累計金額的第一期。
- 在 [期末] 設定計算累計金額的最後一期。
- [支付日期] 若為期初是 1，期末是 0。省略時，預設為 0。

範例 1 計算從期初到各還款次數為止累計攤還的利息

	A	B	C	D	E
1	償還貸款			次數	累計攤還利息
2	借款	\$200,000		1	\$-667
3	利率 (年)	4.0%		2	\$-1,279
4	期間 (月)	12		3	\$-1,836
5	每月還款金額	\$-17,030		4	\$-2,339
6				5	\$-2,786
7				6	\$-3,178
8				7	\$-3,515
9				8	\$-3,796
10				9	\$-4,021
11				10	\$-4,190
12				11	\$-4,303
13				12	\$-4,360
14					
15					

說明 假設年利率 4 % (B3/12)、還款期間 12 個月(B4)、借款 200,000 元(B2) 時，計算從第一次開始，到指定還款次數 (D2) 內累計攤還的利息。定期還款以月為單位，所以年利率要除以 12，變成月利率。

公式 = CUMIPMT(\$B\$3/12,\$B\$4,\$B\$2,1,D2,0)

財務

儲蓄、償還貸款

365

2019

2016

2013

NPV

計算達成目標金額的儲蓄次數或還款次數

傳回固定利率與期間，以本金平均攤還定期償還貸款或零存整付儲蓄時，達到目標金額為止的付款次數。

格式：**NPV(利率, 定期支付額, 現值, [未來值], [支付日期])**

- 在 [利率] 設定利率。
- 在 [定期支付額] 設定每次支付的金額(儲蓄額)。以負值設定支付額。
- [現值] 若是貸款，設定為借款金額；如果是儲蓄，則設定為頭期款。
- [未來值] 若為貸款，設定支付後的餘額(還清時為 0)；如果是儲蓄，設定最終的目標金額。省略時，預設為 0。
- [支付日期] 若為期初是 1，期末是 0。省略時，預設為 0。

範例 1 計算還完借款的還款次數

	A	B	C
1	償還貸款		
2	借款	\$200,000	
3	利率 (年)	4.0%	
4	月支付額	\$-30,000	
5	還款次數	6.753087	
6			

說明 假設年利率 4% (B3/12)、借款 200,000 元 (B2)、每月還款金額為 30,000 元 (B4)，計算到還清借款為止的還款次數。定期還款以月為單位，所以年利率要除以 12，變成月利率。

公式 = NPV(B3/12, B4, B2)

財務

儲蓄、償還貸款

365

2019

2016

2013

ISPMT

計算以本金平均攤還償還貸款時攤還的利息

使用本金平均攤還償還貸款時，傳回指定期數的攤還利息。這是為了維持與試算表軟體 LOTUS1-2-3 的相容性而準備的函數。

格式：**ISPMT(利率, 期, 期間, 現值)**

- 在 [利率] 設定利率。若為每月支付，設定年利率 ÷ 12。
- 在 [期]，假設初期為 0，以介於 [期間]-1 的範圍設定是第幾次付款。
- 在 [期間] 設定總支付次數。若為每月支付，設定為年數 × 12。
- 在 [現值] 設定借款金額。

RATE

計算儲蓄或貸款還款的利率

計算固定期間內投資(貸款或儲蓄)的利率。例如可以計算貸款 50 萬元，一年內每月還款 45,000 元時的貸款利息。

格式：**RATE(期間, 定期支付額, 現值, [未來值], [支付日期], [猜測值])**

- 在 [期間] 設定總支付次數。若為每月支付，設定為年數 × 12。
- 在 [定期支付額] 設定每次的支付金額(儲蓄額)。以負值設定支付額。
- [現值] 若是貸款，設定為借款金額；如果是儲蓄，則設定為頭期款。
- [未來值] 若為貸款，設定支付後的餘額(還清時為 0)；如果是儲蓄，設定最終的目標金額。省略時，預設為 0。
- [支付日期] 若為期初是 1，期末是 0。省略時，預設為 0。
- 在 [猜測值] 設定利率的猜測值。省略時，預設為 10%。

範例 1 計算儲蓄的利率

	A	B	C
1	儲蓄試算		
2	目標金額	\$3,000,000	
3	儲蓄期間(年)	5	
4	儲蓄金額(月)	-\$45,000	
5	利率(年)	4.2182%	
6			

說明 假設五年(B3*12)每月的儲蓄金額為 45,000 元(B4)，計算頭期款為 0，目標金額為 300 萬元(B2)時的利率。傳回值是月利率，所以乘以 12，變成年利率。

公式 =RATE(B3*12,B4,0,B2)*12

PV

計算現值

傳回以固定利率，定期支付貸款或儲蓄時的現值。若是貸款，可以計算能借貸的金額；若是儲蓄，可以計算頭期款。

格式：**PV(利率, 期間, 定期支付額, [未來值], [支付日期])**

- 在 [利率] 設定整個期間內的固定利率。
- 在 [期間] 設定總支付次數。若為每月支付，設定為年數 × 12。
- 在 [定期支付額] 設定每次支付的金額(儲蓄額)。以負值設定支付額。
- [未來值] 若是貸款，設定支付後的餘額，還清時為 0；若是儲蓄，設定最終的目標金額。省略時，預設為 0。
- [支付日期] 若為期初是 1，期末是 0。省略時，預設為 0。

範例 1 計算可以借到的貸款金額

	A	B
1	房屋貸款的借貸金額	
2	每月還款金額	\$-80,000
3	利率(年)	3%
4	期間(年)	30
5	可能的借貸金額	\$18,975,151
6		

說明 假設年利率 3% (B3/12)，在 30 年間 (B4*12) 內，每月還款 80,000 元 (B2) 時，計算可以借到的貸款金額。由於貸款還清，所以第 4 引數的未來值為 0。

公式 =PV(B3/12,B4*12,B2,0)

FV

計算未來值

傳回以固定利率，定期支付貸款或儲蓄時的未來值。若是貸款，可以計算餘額；若是儲蓄，可以計算領回的金額。

格式：**FV(利率, 期間, 定期支付額, [現值], [支付日期])**

- 在 [利率] 設定整個期間內的固定利率。
- 在 [期間] 設定總支付次數。若為每月支付，設定為年數 × 12。
- 在 [定期支付額] 設定每次支付的金額(儲蓄額)。以負值設定支付額。
- [現值] 若是貸款，設定為借款金額；若是儲蓄，則設定為頭期款。
- [支付日期] 若為期初是 1，期末是 0。省略時，預設為 0。

範例 1 計算零存整付儲蓄期滿領回的金額

	A	B
1	零存整付儲蓄期滿領回的金額	
2	儲蓄金額 (月)	\$-50,000
3	利率 (年)	3.0%
4	儲蓄期間 (年)	5
5	期滿領回的金額	\$3,232,336
6		

說明 假設年利率 3 % (B3/12)，五年 (B4*12) 內每月儲蓄 50,000 元 (B2) 時，計算期滿領回的金額。假設頭期款為 0，所以第 4 引數的現值為 0。

公式 =FV(B3/12,B4*12,B2,0)

FVSCHEDULE

計算利率變動時的投資未來值

計算利率變動時的投資或存款的未來值。

格式： FVSCHEDULE(本金, 利率陣列)

- 在 [本金] 設定投資金額或存款金額。與其他財務函數不同的是，這個函數要以正值設定支付額。
- 在 [利率陣列] 以輸入各期利率的儲存格範圍或陣列常數設定投資期間內的變動利率。例如以第一年 2%、第二年 2.5%、第三年 3% 的變動利率投資 100 萬元時，設定為「=FVSCHEDULE(1000000,{0.02,0.025,0.03})」(※ 詳細內容請參考範例檔)。

RRI

由投資金額及期滿時的目標金額計算利率

從投資期間與投資金額計算期滿領回目標金額的複利利率(等價利率)。RRI 函數的公式為「(未來值 / 現值)^(1/ 期間)-1」。

格式： RRI(期間, 現值, 未來值)

- 在 [期間] 設定投資期間。統一想計算利率的期間與單位。例如期間為五年，計算年利率時設定為 5，計算月利率時，設定為 60(5×12)。
- 在 [現值] 設定投資金額(本金)。與其他財務函數不同，要以正值設定支付額。
- 在 [未來值] 設定期滿時想領回的目標金額。

範例 1 由投資金額與目標金額計算利率

	A	B
1	運用年數	5
2	投資本金	\$800,000
3	目標金額	\$1,000,000
4	等價利率(年)	0.0456396
5		

說明 計算運用年數五年(B1)，投資本金為 80 萬元(B2)，期滿時領回的目標金額是 100 萬元(B3)的等價利率。

公式 =RRI(B1,B2,B3)

PDURATION

計算投資金額達到目標金額所需的時間

利用引數設定現值與利率，計算達到未來值的時間。例如，投資 10 萬元，年利率為 2.5% 時，計算達到 12 萬元的年數 (7.38 年)。

格式： **PDURATION(利率, 現值, 未來值)**

- 在 [利率] 設定投資的利率。
- 在 [現值] 設定現值。
- 在 [未來值] 設定當作目標的未來值。

NPV

計算定期現金流的淨現值

由貼現率及未來各期支出與收入 (現金流) 計算投資的淨現值。淨現值是把未來的收支金額轉換成現值，判斷投資的一項指標。

格式： **NPV(貼現率, 值 1, [值 2], ...)**

- 在 [貼現率] 設定投資期間的固定貼現率。
- 在 [值] 設定代表支出 (負值) 和收入 (正值) 的金額。假設定期在各期的期末發生，設定的順序就會是現金流的順序。

Hint • 最初的現金流在第一期的期初發生時，不可將該現金流設定為引數，必須加入 NPV 函數的計算結果。例如，範例最初的現金流 (A4) 發生在第一期的期初時，設定為「=NPV(B1,B4:D4)+A4」。

• 淨現值為 0 時，即使投資也不會產生獲利。0 以上才有獲利，而且愈大愈好。

範例 1 計算有定期收益時的淨現值

	A	B	C	D
1	年貼現率	10%	淨現值	\$90,499
2				
3	初期投資金額	第1年	第2年	第3年
4	\$-1,000,000	\$350,000	\$400,000	\$600,000

公式 =NPV(B1,A4:D4)

說明 假設投資 100 萬且三年內有定期收益 (A4:D4)，計算年貼現率為 10% (B1) 時的淨現值。